Computational Logic

The (ISO-)Prolog Programming Language
(ISO-)Prolog

- A practical programming language based on the logic programming paradigm.
- Main differences with “pure” logic programming:
  - depth-first search rule, left-to-right control rule,
  - more control of the execution flow,
  - many useful pre-defined predicates (some not declarative, for efficiency),
  - higher-order and meta-logical capabilities, ...
- Advantages:
  - it can be compiled into fast and efficient code (including native code),
  - more expressive power,
  - industry standard (ISO-Prolog),
  - mature implementations with modules, graphical environments, interfaces, ...
- Drawbacks of “classical” systems (and how addressed by modern systems):
  - Depth-first search rule is efficient but can lead to incompleteness → alternative search strategies (e.g., Ciao’s bfall, etc.).
  - No occur check in unification (which led to unsoundness in older systems) → support regular (i.e., infinite) trees: $X = f(X)$ (already constraint-LP).
Programming Interface (Writing and Running Programs)

- **Not** specified in the ISO-Prolog language standard.
  → Is left to each particular system implementing the standard.

- This typically includes issues such as:
  ◦ User interaction (top-level, GUI, etc.).
  ◦ Interpreter(s).
  ◦ Compiler(s).
  ◦ Debugger(s).
  ◦ *(Module system.)*

- Different Prolog systems offer different facilities for these purposes.

- **See the part on Developing Programs with a Logic Programming System for more details for the particular system used in the course (Ciao).**
Comparison with Imperative and Functional Languages

- **Programs without search** (that do not perform “deep” backtracking):
  - Generally (if no disjunction etc. used) this means programs that:
    * Have only one clause per procedure, or
    * if several clauses, only one of them unifies for every call to that predicate.
  - Note that this is dependent on call mode, i.e., which variables are bound on a given call.
  - Because of the left-to-right rule, these programs run in Prolog similarly to their imperative and (strict) functional counterparts.
  - Imperative/functional programs can be directly expressed as such programs.

- **Programs with search** (perform “deep” backtracking):
  - These are programs that have at least one procedure that:
    * has multiple clauses, and
    * more than one of them unifies for some calls to that procedure.
  - Again, this is dependent on call mode.
  - These programs perform search (backtracking-based, or other search rules).
  - They have no direct counterparts in imperative or functional programming.
Comparison with Imperative and Functional Languages (Contd.)

- Conventional languages and Prolog both implement *(forward) continuations*: the place to go after a procedure call *succeeds*. I.e., in:

\[
\begin{align*}
p(X, Y) & : - q(X, Z), r(Z, Y). \\
q(X, Z) & : - \ldots
\end{align*}
\]

when the procedure call to \texttt{q/2} finishes (with “success”), execution continues in \texttt{p/2}, just after the call to \texttt{q/2}, i.e., at the call to \texttt{r/2} (the *forward continuation*).

- In Prolog, *when there are procedures with multiple definitions*, there is also a *backward continuation*: the place to go to if there is a *failure*. I.e., in:

\[
\begin{align*}
p(X, Y) & : - q(X, Z), r(Z, Y). \\
p(X, Y) & : - \ldots \\
q(X, Z) & : - \ldots
\end{align*}
\]

if the call to \texttt{q/2} succeeds, it is as above, but if it fails, execution continues at (“backtracks to”) the *previous alternative*: the second clause of \texttt{p/2} (the *backward continuation*).

- We say that \texttt{p/2} has a *choice point*.

- Again, the debugger (see later) can be useful to observe how execution proceeds.
Control of Search in Prolog

Again, conventional programs (no search) execute conventionally.

Programs with search: programmer has at least three ways of controlling execution:

1. The ordering of literals in the body of a clause:
   - Profound effect on the size of the computation (in the limit, on termination).
     Compare executing \( p(X), q(X, Y) \) with executing \( q(X, Y), p(X) \) in:

     \[
     \begin{align*}
     p(X) & : - X = 4. \\
     p(X) & : - X = 5. \\
     q(X, Y) & : - X = 1, Y = a, \ldots \\
     q(X, Y) & : - X = 2, Y = b, \ldots \\
     q(X, Y) & : - X = 4, Y = c, \ldots \\
     q(X, Y) & : - X = 4, Y = d, \ldots
     \end{align*}
     \]

     \( p(X), q(X, Y) \) is more efficient: execution of \( p/2 \) reduces the choices of \( q/2 \).

   - Note that optimal order depends on the variable instantiation mode:
     E.g., if in \( q(X, d), p(X) \), this order is better than \( p(X), q(X, d) \).
The ordering of clauses in a predicate:

- Affects the order in which solutions are generated.
  
  E.g., in the previous example we get:
  
  \[ X=4, Y=c \] as the first solution and \[ X=4, Y=d \] as the second.

  If we reorder \( q/2 \):
  
  \[
  \begin{align*}
  p(X) & :- X = 4. \\
  p(X) & :- X = 5. \\
  q(X, Y) & :- X = 4, Y = d, \ldots \\
  q(X, Y) & :- X = 4, Y = c. \ldots \\
  q(X, Y) & :- X = 2, Y = b, \ldots \\
  q(X, Y) & :- X = 1, Y = a, \ldots
  \end{align*}
  \]

  we get \[ X=4, Y=d \] first and then \[ X=4, Y=c \].

- If a subset of the solutions is requested, then clause order affects:
  
  ◇ the size of the computation,
  
  ◇ and, at the limit, termination!

  Else, little significance unless computation is infinite and/or pruning is used.

The pruning operators (e.g., “cut”), which cut choices dynamically —see later.
The ISO Standard (Overview)

- Syntax (incl. operators) and operational semantics
- Arithmetic
- Checking basic types and state
- Structure inspection and term comparison
- Input/Output
- Pruning operators (cut)
- Meta-calls, higher-order, aggregation predicates
- Negation as failure, cut-fail
- Dynamic program modification
- Meta-interpreters
- Incomplete data structures
- Exception handling

Additionally (not in standard):
- Definite Clause Grammars (DCGs): parsing
Variables and constants as before:

- Variables (start with capital or _): X, Value, A, A1, _3, _result.
- Constants (start w/small letter or in ’ ’): x, =, [], ’Algol-3’, ’Don’t’.

Note: in Prolog terminology constants are also referred to as “atom”s.

Numbers: 0, 999, -77, 5.23, 0.23e-5, 0.23E-5.

Strings (of “codes”): "Prolog" = [80, 114, 111, 108, 111, 103] (list of ASCII character codes).

- Note: if ?- set_prolog_flag(write_strings, on).
  character lists are printed as strings: " ".

Comments:

- Using %: rest of line is a comment.
- Using /* ... */: everything in between is a comment.
Prolog Syntax — Defining Operators

- Certain functors and predicate symbols are predefined as infix, prefix, or postfix operators, aside from the standard term notation.
- Very useful to make programs (or data files) more readable.
- Stated using *operator declarations*:
  
  ```prolog
  :- op(< precedence >, < type >, < operator(s) >).
  ```

  where:
  
  - < precedence >: is an integer from 1 to 1200.
    E.g., if ‘+’ has higher precedence than ‘/’, then
    \[ a + b / c \equiv a + (b / c) \equiv + (a, /(b, c)) \].
    Otherwise, use parenthesis: \[ / (+ (a, b), c) \equiv (a + b) / c \].
  
  - < type >:
    * infix: \texttt{xfx} (not associative), \texttt{xfy} (right associative), \texttt{yfx} (left associative).
    * prefix: \texttt{fx} (non-associative), \texttt{fy} (associative).
    * postfix: \texttt{xf} (non-associative), \texttt{yf} (associative).
  
  - < operator(s) >: can be a single atom or a list of atoms.
Prolog Syntax — Operators (Contd.)

- Examples:

<table>
<thead>
<tr>
<th>Standard Notation</th>
<th>Operator Notation</th>
</tr>
</thead>
<tbody>
<tr>
<td>`'+'(a,'/'(b,c))</td>
<td>a+b/c</td>
</tr>
<tr>
<td><code>is(X, mod(34, 7))</code></td>
<td>X is 34 mod 7</td>
</tr>
<tr>
<td><code>'&lt;( '+'(3,4),8)</code></td>
<td>3+4 &lt; 8</td>
</tr>
<tr>
<td><code>='(X,f(Y))</code></td>
<td>X = f(Y)</td>
</tr>
<tr>
<td><code>'-(3)</code></td>
<td>-3</td>
</tr>
<tr>
<td><code>spy('/(foo,3))</code></td>
<td>spy foo/3</td>
</tr>
<tr>
<td><code>':-'(p(X),q(Y))</code></td>
<td>p(X) :- q(Y)</td>
</tr>
<tr>
<td><code>':-'(p(X),','(q(Y),r(Z)))</code></td>
<td>p(X) :- q(Y),r(Z)</td>
</tr>
</tbody>
</table>

- Note that, with this syntax convention, Prolog clauses are also Prolog terms!

- Parenthesis must always be used for operators with higher priority than 1000 (i.e., the priority of `',')`:

  ```prolog
  ... assert( (p :- q)), ...
  ```

- Operators are local to modules (explained later).
• Typical standard operators:

```prolog
:- op( 1200, xfx, [ :-, --> ]).
:- op( 1200, fx, [ :-, ?- ]).
:- op( 1150, fx, [ mode, public, dynamic,
                  multifile, block, meta_predicate,
                  parallel, sequential ]).
:- op( 1100, xfy, [ ; ]).
:- op( 1050, xfy, [ -> ]).
:- op( 1000, xfy, [ ',' ]).
:- op( 900, fy, [ \+, spy, nospy ]).
:- op( 700, xfx, [ =, is, =.., ==, \==, @<, @>, @=<, @>=,
                  =:=, =\=, <, >, =<, >= ]).
:- op( 550, xfy, [ : ]).
:- op( 500, yfx, [ +, -, #, \/, \//, \\ ]).
:- op( 500, fx, [ +, - ]).
:- op( 400, yfx, [ *, /, //, <<, >> ]).
:- op( 300, xfx, [ mod ]).
:- op( 200, xfy, [ ^ ]).
```
The Execution Mechanism of (classical) Prolog

- Always execute calls in the body of clauses *left-to-right*.
- When entering a procedure, if several clauses unify (a *choice point*), take the *first unifying clause* (i.e., the leftmost unexplored branch).
- On failure, *backtrack* to the *next unexplored clause* of the *last choice point*.

\[
\text{grandparent}(C,G) :- \text{parent}(C,P), \text{parent}(P,G).
\]

\[
\text{parent}(C,P) :- \text{father}(C,P).
\]

\[
\text{parent}(C,P) :- \text{mother}(C,P).
\]

\[
\text{father}(\text{charles}, \text{philip}).
\]

\[
\text{father}(\text{ana}, \text{george}).
\]

\[
\text{mother}(\text{charles}, \text{ana}).
\]

- Check how Prolog explores this tree by running the *debugger*!
Built-in Arithmetic

- Practicality: interface to the underlying CPU arithmetic capabilities.
- These arithmetic operations are not as general as their logical counterparts.
- Interface: evaluator of arithmetic terms.
- The type of arithmetic terms (arithexpr/1):
  - a number is an arithmetic term,
  - if $f$ is an $n$-ary arithmetic functor and $X_1, \ldots, X_n$ are arithmetic terms then $f(X_1, \ldots, X_n)$ is an arithmetic term.
- Arithmetic functors: $+, -, *, /$ (float quotient), $//$ (integer quotient), $\text{mod}$, ...
  (see later).

Examples:
  - $(3*X+Y)/Z$, correct if when evaluated $X$, $Y$ and $Z$ are arithmetic terms, otherwise it will raise an error.
  - $a+3*X$ raises an error (because $a$ is not an arithmetic term).
Built-in Arithmetic (Contd.) – The arithexpr type

arithexpr :=
  | num
  | - arithexpr
  | -- arithexpr
  | aritexpr + aritexpr
  | aritexpr * aritexpr
  | aritexpr / aritexpr
  | aritexpr ** aritexpr
  | exp(arithexpr)
  | log(arithexpr)
  | sin(arithexpr)
  | atan(arithexpr)
  | abs(arithexpr)
  | sign(arithexpr)
  | float_integer_part(arithexpr)
  | float_fractional_part(arithexpr)
  | float(arithexpr)
  | 

Other operators (see manuals):

- rem, mod, gcd, >>, <<, /\, \/, \/, #, ...
- integer, truncate, floor, round, ceiling, ...
Built-in Arithmetic (Contd.)

- Built-in arithmetic predicates:
  - the usual `<`, `>`, `=<`, `>=`, `=:=` (arithmetic equal), `===` (arithmetic not equal), ...
    - Both arguments are evaluated and their results are compared.
  - Z is X
    - X (which must be an arithmetic term) is evaluated and result is unified with Z.

- Examples:
  - X is 3+3//2.
  - Z = 3//2, X is 3+Z.

- Examples of failure and errors:
  - X=3, Y=4, Y<X+1.
  - X=3, Y=4, X is Y+1.
  - X=3, Y=4, X =:= Y.
  - Y=4, Y<a+1.
  - X is Z+1.
  - X=3, X =:= f(a).

 fail (the system will backtrack).
 fail (the system will backtrack).
 fail (the system will backtrack).
 throws error (the system will abort).
 throws error (the system will abort).
 throws error (the system will abort).
Arithmetic Programs

- \texttt{plus(X,Y,Z):- Z is X + Y}
  - Only works in one direction (X and Y bound to arithmetic terms).
  - Meta-logical tests (see later) allow using it in both directions.
  - We have lost the recursive structure of the numbers.
  - But we have won (a lot) in performance!

- Factorial:
  Using Peano arithmetic:
  \begin{verbatim}
  factorial(0, s(0)).
  factorial(s(N), F) :-
      factorial(N, F1),
      times(s(N), F1, F).
  \end{verbatim}
  Using Prolog arithmetic:
  \begin{verbatim}
  factorial(0, 1).
  factorial(N, F) :-
      N > 0,
      N1 \textit{is} N-1,
      factorial(N1, F1),
      F \textit{is} F1*N.
  \end{verbatim}

- Wrong goal order can raise an error (e.g., moving last call to \texttt{is/2} before call to \texttt{factorial}).
Dynamic Checking of Basic Types

- Unary relations which check the type of a term:
  - `integer(X)`
  - `float(X)`
  - `number(X)`
  - `atom(X)` (nonvariable term of arity 0 other than a number)
  - `atomic(X)` atom or number
  - ...

- They behave as if defined by a (possibly infinite) table of facts (in part, see below).
- They either succeed or fail, but do not produce an error.
- Thus, they cannot be used to *generate* (e.g., if argument is a variable, they fail instead of instantiating it to possible values).
- This behaviour is outside first order logic because it allows checking the instantiation state of a variable.
Example: implementing a better behavior for plus/3:

\[
\begin{align*}
\text{plus}(X,Y,Z) &: \text{number}(X), \text{number}(Y), \ Z \text{ is } X + Y. \\
\text{plus}(X,Y,Z) &: \text{number}(X), \text{number}(Z), \ Y \text{ is } Z - X. \\
\text{plus}(X,Y,Z) &: \text{number}(Y), \text{number}(Z), \ X \text{ is } Z - Y.
\end{align*}
\]

Then:

?- plus(3,Y,5).
Y = 2 ?

Still, it cannot be used to partition a number into two others:

?- plus(X,Y,5).
no

(in fact, this should raise an error, rather than simply failing).
Structure Inspection

- **functor**\((X, F, A)\):

  - \(X\) is a compound term \(f(X_1, \ldots, X_n)\) → \(F = f\) \(A = n\)
  - \(F\) is the atom \(f\) and \(A\) is the integer \(n\) → \(X = f(X_1, \ldots, X_n)\)
  - Error if \(x\), and either \(F\) or \(A\) are variables
  - Fails if the unification fails, \(A\) is not an integer, or \(F\) is not an atom

Examples:

- **functor**\((t(b,a),F,A)\) → \(F = t\), \(A = 2\).
- **functor**\((Term,f,3)\) → \(Term = f(_, _, _)\).
- **functor**\((Vector,v,100)\) → \(Vector = v(_, \ldots, _)\).

(Note: in some systems functor arity is limited to 256)
• \texttt{arg(N, X, Arg)}:
  ◦ \texttt{N} integer, \texttt{X} compound term → \texttt{Arg} unified with \texttt{n-th} argument of \texttt{X}.
  ◦ Allows accessing a structure argument in constant time and in a compact way.
  ◦ Error if \texttt{N} is not an integer, or if \texttt{X} is a free variable.
  ◦ Fails if the unification fails.

Examples:
\begin{verbatim}
?- _T=date(9,February,1947), \texttt{arg}(3,_T,X).
X = 1947
?- _T=date(9,February,1947), _T=date(_,_,X).
X = 1947
?- functor(Array,array,5),
   \texttt{arg}(1,Array,black),
   \texttt{arg}(5,Array,white).
Array = array(black,_,_,_,white).
\end{verbatim}

• What does \texttt{?- arg(2,[a,b,c,d],X).} return?
Example of Structure Inspection

- Define `subterm(Sub,Term)` (Term will always be a compound term):

```prolog
subterm(Term,Term).
subterm(Sub,Term):-
    functor(Term,F,N),
    subterm(N,Sub,Term).

subterm(N,Sub,Term):-
    arg(N,Term,Arg), % also checks N > 0 (arg/1 fails otherwise!)
    subterm(Sub,Arg).
subterm(N,Sub,Term):-
    N>1,
    N1 is N-1,
    subterm(N1,Sub,Term).
```
Example of Structure Access

- Define `add_arrays(A1,A2,A3)`: 

```
add_arrays(A1,A2,A3):- % Same N imposes equal length:
    functor(A1,array,N), functor(A2,array,N), functor(A3,array,N),
    add_elements(N,A1,A2,A3).
```

```
add_elements(0,_A1,_A2,_A3).
add_elements(I,A1,A2,A3):-
    I>0, arg(I,A1,X1), arg(I,A2,X2), arg(I,A3,X3),
    X3 is X1 + X2, I1 is I - 1,
    add_elements(I1,A1,A2,A3).
```

- Alternative, using lists instead of structures:

```
add_arrays_lists([],[],[]).
add_arrays_lists([X|Xs],[Y|Ys],[Z|Zs]):-
    Z is X + Y,
    add_arrays_lists(Xs,Ys,Zs).
```

- In the latter case, where do we check that the three lists are of equal length?
Higher-Order Structure Inspection

- $T = \text{univ}L$ (known as “univ”)
  - $L$ is the decomposition of a term $T$ into a list comprising its principal functor followed by its arguments.

```prolog
?- date(9, february, 1947) =.. L.
L = [date,9,february,1947].
?- _F = '+', X =.. [F,a,b].
X = a + b.
```

- Allows *implementing* higher-order primitives (see later).
  Example: Extending derivative

```prolog
derivative(sin(X),X,cos(X)).
derivative(cos(X),X,-sin(X)).
derivative(FG_X, X, DF_G * DG_X):-
    FG_X =.. [_, G_X],
    derivative(FG_X, G_X, DF_G), derivative(G_X, X, DG_X).
```

- But *do not use* unless strictly necessary: expensive in time and memory.
Conversion Between Strings and Atoms (New Atom Creation)

- Classical primitive: \texttt{name}(A,S)
  - A is the atom/number whose name is the list of ASCII characters S
  
  \begin{verbatim}
  \texttt{?- name(hello,S).}
  A = hello
  \texttt{?- name(A,"hello").}
  A = hello
  \end{verbatim}

- Ambiguity when converting strings which represent numbers.
  Example: \texttt{?- name(’1’,X), name(Y,X).}

- In the ISO standard fixed by dividing into two:
  * \texttt{atom_codes(Atom,String)}
  * \texttt{number_codes(Number,String)}
Meta-Logical Predicates

- **var**($X$): succeed iff $X$ is a free variable.
  
  ?- var(X), X = f(a). %Succeeds  
  ?- X = f(a), var(X). %Fails

- **nonvar**($X$): succeed iff $X$ is not a free variable.
  
  ?- X = f(Y), nonvar(X). %Succeeds

- **ground**($X$): succeed iff $X$ is fully instantiated.
  
  ?- X = f(Y), ground(X). %Fails

- Outside the scope of first order logic.
- Uses:
  - control goal order,
  - restore some flexibility to programs using certain builtins.
Meta-Logical Predicates (Contd.) – choosing implementations

- Example: list length:

  \[\text{length}([], 0)\].
  \[\text{length}([\_|T], N) :- \text{length}(T, TN), N \text{ is } TN+1.\]

- Choosing between two implementations based on calling mode. I.e., implementing *reversibility* "by hand."

  \[\text{length}(L, N) :- \text{var}(L), \text{integer}(N), \text{create_list}(N, L).\]
  \[\text{length}(L, N) :- \text{nonvar}(L), \text{compute_length}(L, N).\]

  \[\text{create_list}(0, []).\]
  \[\text{create_list}(N, [\_|T]) :- N > 0, NT \text{ is } N-1, \text{create_list}(NT, T).\]

  \[\text{compute_length}([], 0)\].
  \[\text{compute_length}([\_|T], N) :- \text{compute_length}(T, TN), N \text{ is } TN+1.\]

- Not really needed: the normal definition of length is actually reversible!

- Although reversing traditional list length is less efficient than length_num(N,L) when L is a variable and N a number.
Example (Contd): Choosing between implementations based on calling mode.

With more efficient version of `compute_length/2` (using an “acummmulating parameter” —see slides on Prolog efficency):

```prolog
length(L,N):- var(L), integer(N), create_list(N,L).
length(L,N):- nonvar(L), compute_length(L,N).
create_list(0,[]).
create_list(N,[_|T]):- N > 0, NT is N-1, create_list(NT,T).
compute_length(L,N) :- compute_length_(L,0,N).
compute_length_([],N,N).
compute_length_([_|T],A,N) :- NA is A+1,
    compute_length_(T,NA,N).
```

Comparing Non-ground Terms

- Many applications need comparisons between non-ground/non-numeric terms.

- Identity tests:
  - $X \equiv Y$ (identical)
  - $X \not\equiv Y$ (not identical)

  $$\text{?- } f(X) \equiv f(X). \% \text{ Succeeds}$$

  $$\text{?- } f(X) \equiv f(Y). \% \text{ Fails}$$

- Term ordering:
  - $X @ Y$, $X \geq Y$, $X @< Y$, $X @< Y$ (alphabetic/lexicographic order)

  $$\text{?- } f(a) @> f(b). \% \text{ Fails}$$

  $$\text{?- } f(b) @> f(a). \% \text{ Succeeds}$$

  $$\text{?- } f(X) @> f(Y). \% \text{ Implementation dependent!}$$
Comparing Non-ground Terms (Contd.)

- Reconsider `subterm/2` with non-ground terms

```prolog
subterm(Sub, Term) :- Sub == Term.
subterm(Sub, Term) :- nonvar(Term), functor(Term, F, N), subterm(N, Sub, Term).
```

where `subterm/3` is identical to the previous definition

- Insert an item into an ordered list:

```prolog
insert([], Item, [Item]).
insert([H|T], Item, [H|T]) :- H == Item.
insert([H|T], Item, [Item, H|T]) :- H @> Item.
insert([H|T], Item, [H|NewT]) :- H @< Item, insert(T, Item, NewT).
```

- Compare with the same program with the second clause defined as

```prolog
insert([H|T], Item, [Item|T]) :- H = Item.
```
A minimal set of input-output predicates ("DEC-10 Prolog I/O"):  

<table>
<thead>
<tr>
<th>Class</th>
<th>Predicate</th>
<th>Explanation</th>
</tr>
</thead>
<tbody>
<tr>
<td>I/O stream control</td>
<td>see(File)</td>
<td>File becomes the current input stream.</td>
</tr>
<tr>
<td></td>
<td>seeing(File)</td>
<td>The current input stream is File.</td>
</tr>
<tr>
<td></td>
<td>seen</td>
<td>Close the current input stream.</td>
</tr>
<tr>
<td></td>
<td>tell(File)</td>
<td>File becomes the current output stream.</td>
</tr>
<tr>
<td></td>
<td>telling(File)</td>
<td>The current output stream is File.</td>
</tr>
<tr>
<td></td>
<td>told</td>
<td>Close the current output stream.</td>
</tr>
<tr>
<td>Term I/O</td>
<td>write(X)</td>
<td>Write the term X on the current output stream.</td>
</tr>
<tr>
<td></td>
<td>nl</td>
<td>Start a new line on the current output stream.</td>
</tr>
<tr>
<td></td>
<td>read(X)</td>
<td>Read a term (finished by a full stop) from the current input stream and unify it with X.</td>
</tr>
<tr>
<td>Character I/O</td>
<td>put_code(N)</td>
<td>Write the ASCII character code N. N can be a string of length one.</td>
</tr>
<tr>
<td></td>
<td>get_code(N)</td>
<td>Read the next character code and unify its ASCII code with N.</td>
</tr>
</tbody>
</table>
Other stream-based input-output predicates:

<table>
<thead>
<tr>
<th>Class</th>
<th>Predicate</th>
<th>Explanation</th>
</tr>
</thead>
<tbody>
<tr>
<td>I/O stream control</td>
<td><code>open(File,M,S)</code></td>
<td>Open File with mode M and return in S the stream associated with the file. M may be <code>read</code>, <code>write</code> or <code>append</code>. Close the stream ‘Stream’.</td>
</tr>
<tr>
<td></td>
<td><code>close(S)</code></td>
<td></td>
</tr>
<tr>
<td>Term I/O</td>
<td><code>write(S,X)</code></td>
<td>Write the term X on stream S.</td>
</tr>
<tr>
<td></td>
<td><code>nl(S)</code></td>
<td>Start a new line on stream S.</td>
</tr>
<tr>
<td></td>
<td><code>read(S,X)</code></td>
<td>Read a term (finished by a full stop) from the stream S and unify it with X.</td>
</tr>
<tr>
<td>Character I/O</td>
<td><code>put_code(S,N)</code></td>
<td>Write the ASCII character code N on stream S.</td>
</tr>
<tr>
<td></td>
<td><code>get_code(S,N)</code></td>
<td>Read from stream S the next character code and unify its ASCII code with N.</td>
</tr>
</tbody>
</table>
• Example:

```prolog
write_list_to_file(L,F) :-
    telling(OldOutput), % Grab current output stream.
    tell(F), write_list(L), % Write into F.
    told, % Close.
    tell(OldOutput). % Reset previous output stream.
```

```prolog
write_list([]).
write_list([X|Xs]) :- write(X), nl, write_list(Xs).
```

• More powerful and format-based input-output predicates are available (see, e.g., `format/2` and `format/3` – Prolog system manuals).

• All these input-output predicates are “side-effects”!
Pruning Operator: Cut

- A “cut” (predicate !/0) commits Prolog to all the choices made since the parent goal was unified with the head of the clause in which the cut appears.

- Thus, it prunes:
  - all clauses below the clause in which the cut appears, and
  - all alternative solutions to the goals in the clause to the left of the cut.

But it does not affect the search in the goals to the right of the cut.

```plaintext
s(1). p(X, Y):- l(X), ...                  r(8).
s(2). p(X, Y):- r(X), !, ...              r(9).
      p(X, Y):- m(X), ...
```

with query `?- s(A), p(B, C)`.

If execution reaches the cut (`!`):

- The second alternative of `r/1` is not considered.
- The third clause of `p/2` is not considered.
Pruning Operator: Cut (Contd.)

s(1).
\[
p(X,Y) :- \ l(X), \ldots \quad r(8).
\]
s(2).
\[
p(X,Y) :- \ r(X),!,\ldots \quad r(9).
p(X,Y) :- \ m(X),\ldots
\]
“Types” of Cut

- **White** cuts: do not discard solutions.

  ```prolog
data(X,Y,X) :- X > Y, !.
data(X,Y) :- X =< Y.
```

They affect neither completeness nor correctness – use them freely.
(In many cases the system “introduces” them automatically.)

- **Green** cuts: discard correct solutions which are not needed.

  ```prolog
address(X,Add) :- home_address(X,Add), !.
address(X,Add) :- business_address(X,Add).
membercheck(X,[X|Xs]) :- !.
membercheck(X,[Y|Xs]) :- membercheck(X,Xs).
```

They affect completeness but not correctness.
Necessary in many situations (but beware!).
“Types” of Cut (Contd.)

- Red cuts: discard solutions which are not correct according to the intended meaning.

  ◇ Example:
  
  \[
  \text{max}(X,Y,X) :- X > Y, !. \\
  \text{max}(X,Y,Y).
  \]

  *wrong answers to, e.g., \(?- \text{max}(5, 2, 2).*

  ◇ Example:
  
  \[
  \text{days_in_year}(X,366) :- \text{leap_year}(X), !. \\
  \text{days_in_year}(X,365).
  \]

  *wrong answers to, e.g., \(?- \text{days_in_year}(a, D).*

Red cuts affect completeness and one can no longer rely on the strict declarative interpretation of the program for reasoning about correctness – avoid when possible.
Meta-calls and Implementing Higher Order

- The meta-call \texttt{call(X)} converts a term \( X \) into a goal and calls it.
- When called, \( X \) must be instantiated to a term, otherwise an error is reported.
- Used for meta-programming, specially interpreters and shells. Also for defining negation (as we will see) and \textit{implementing} higher order.

- Example:
  \begin{verbatim}
  q(a). p(X) :- call(X).
  ?- p(q(Y)).
  Y = a
  \end{verbatim}

- Example:
  \begin{verbatim}
  q(a,b). apply(F,Args) :- G =.. [F|Args], call(G).
  ?- apply(q,[Y,Z]).
  Y = a
  Z = b
  \end{verbatim}
Meta–calls – Aggregation Predicates

- Other meta-calls are, e.g., `findall/3`, `bagof/3`, and `setof/3`.

- **findall**(*Term*, *Goal*, *ListResults*): *ListResults* is the set of all instances of *Term* such that *Goal* is satisfied
  - If there are no instances of *Term*, *ListResults* is `[]`
  - For termination, the number of solutions should be finite (and enumerable in finite time).

\[
\text{likes}(\text{bill}, \text{cider}). \\
\text{likes}(\text{dick}, \text{beer}). \\
\text{likes}(\text{tom}, \text{beer}). \\
\text{likes}(\text{tom}, \text{cider}). \\
\text{likes}(\text{harry}, \text{beer}). \\
\text{likes}(\text{jan}, \text{cider}).
\]

?- **findall**(X, likes(X,Y), S).
S = [bill,dick,tom,tom,harry,jan] ?
yes

?- **findall**(X, likes(X,water), S).
S = [] ?
yes

?-
• **setof**(Term, Goal, ListResults): ListResults is the ordered set (no duplicates) of all instances of Term such that Goal is satisfied

  ◦ If there are no instances of Term the predicate fails
  ◦ The set should be finite (and enumerable in finite time)
  ◦ If there are un-instantiated variables in Goal which do not also appear in Term then a call to this built-in predicate may backtrack, generating alternative values for ListResults corresponding to different instantiations of the free variables of Goal
  ◦ Variables in Goal will not be treated as free if they are explicitly bound within Goal by an existential quantifier as in \( \exists Y^\ldots \) (then, they behave as in **findall**/3)

• **bagof**/3:
  
same, but returns list unsorted and with duplicates (in backtracking order).
Meta-calls – Aggregation Predicates: Examples

likes(bill, cider).
likes(dick, beer).
likes(harry, beer).
likes(jan, cider).
likes(tom, beer).
likes(tom, cider).

?- setof(X, likes(X,Y), S).
S = [dick, harry, tom],
Y = beer ? ;
S = [bill, jan, tom],
Y = cider ? ;
no

?- setof((Y,S), setof(X, likes(X,Y), S), SS).
SS = [(beer, [dick, harry, tom]),
      (cider, [bill, jan, tom])] ? ;
no

?- setof(X, Y^(likes(X,Y)), S).
S = [bill, dick, harry, jan, tom] ? ;
no
Meta-calls – Negation as Failure

- Uses the meta-call facilities, the cut and a system predicate `fail` that fails when executed (similar to calling `a=b`).

\[
\text{not}(\text{Goal}) :- \text{call}(\text{Goal}), !, \text{fail}.
\]

- Available as the (prefix) predicate `\+)/1: `+ member(c, [a,k,l])`

- It will never instantiate variables.

- Termination of `\text{not}(\text{Goal})` depends on termination of `\text{Goal}`. `\text{not}(\text{Goal})` will terminate if a success node for `\text{Goal}` is found before an infinite branch.

- It is very useful but dangerous:

\[
\text{unmarried\_student}(X) :- \text{not}(\text{married}(X)), \text{student}(X).
\]

\[
\text{student}(\text{joe}).
\]

\[
\text{married}(\text{john}).
\]

- Works correctly for **ground goals** (programmer’s responsibility to ensure this).
Cut-Fail

- Cut-fail combinations allow forcing the failure of a predicate — somehow specifying a negative answer (useful but very dangerous!).

- Example – testing groundness: fail as soon as a free variable is found.

```prolog
ground(Term) :- var(Term), !, fail.
ground(Term) :-
    nonvar(Term),
    functor(Term,F,N),
    ground(N,Term).

ground(0,T).  % All subterms traversed
ground(N,T):-
    N>0,
    arg(N,T,Arg),
    ground(Arg),
    N1 is N-1,
    ground(N1,T).
```
Dynamic Program Modification (I)

• assert/1, retract/1, abolish/1, ...

• Very powerful: allows run–time modification of programs. Can also be used to simulate global variables.

• Sometimes this is very useful, but very often a mistake:
  ◦ Code hard to read, hard to understand, hard to debug.
  ◦ Typically, slow.

• Program modification has to be used scarcely, carefully, locally.

• Still, assertion and retraction can be logically justified in some cases:
  ◦ Assertion of clauses which logically follow from the program. (lemmas)
  ◦ Retraction of clauses which are logically redundant.

• Other typically non-harmful use: simple global switches.

• Behavior/requirements may differ between Prolog implementations. Typically, the predicate must be declared :- dynamic.
Dynamic Program Modification (II)

- Example program:
  ```prolog
  relate_numbers(X, Y):- assert(related(X, Y)).
  unrelate_numbers(X, Y):- retract(related(X, Y)).
  ```

- Example query:
  ```prolog
  ?- related(1, 2).
  {EXISTENCE ERROR: ...}
  ?- relate_numbers(1, 2).
  yes
  ?- related(1, 2).
  yes
  ?- unrelate_numbers(1, 2).
  yes
  ?- related(1, 2).
  no
  ```

- Rules can be asserted dynamically as well.
Dynamic Program Modification (III)

- Example program:

```prolog
fib(0, 0).
fib(1, 1).
fib(N, F):- 
    N > 1, 
    N1 is N - 1, 
    N2 is N1 - 1, 
    fib(N1, F1), 
    fib(N2, F2), 
    F is F1 + F2.
```

```prolog
lfib(N, F):- lemma_fib(N, F), !.
lfib(N, F):- 
    N > 1, 
    N1 is N - 1, 
    N2 is N1 - 1, 
    lfib(N1, F1), 
    lfib(N2, F2), 
    F is F1 + F2, 
    assert(lemma_fib(N, F)). 
:- dynamic lemma_fib/2.
lemma_fib(0, 0).
lemma_fib(1, 1).
```

- Compare `fib(24, N)` versus `lfib(24, N)`.
Meta-Interpreters

- **clause(\textless head\textgreater , \textless body\textgreater)\)**
  - Reads a clause \textit{head} \texttt{:-} \textit{body} from the program.
  - For facts \textit{body} is \texttt{true}.
- To use \texttt{clause/2} a predicate must be declared \texttt{dynamic}.
- Simple (“vanilla”) meta-interpreter:
  ```prolog
  solve(\texttt{true}).
solve((A,B)) :- solve(A), solve(B).
solve(A) :- clause(A,B), solve(B).
  ```
- This code can be enhanced to do many tasks: tracing, debugging, explanations in expert systems, implementing other computation rules, ...
- Issues / interactions with module system.
Incomplete Data Structures

• Example – difference lists:

  ◦ A pseudo-type:

  \[
  \text{dlist}(X-Y) :- \var(X), \!, X==Y.
  \]
  \[
  \text{dlist}([_|DL]-X) :- \text{dlist}(DL-X).
  \]

  (Note: just for “minimal” difference lists, and not declarative because of \(==/2\))

  ◦ Allows us to keep a pointer to the end of the list.

  ◦ Allows appending in constant time:

  \[
  \text{append_dl}(B1-E1,B2-E2,B3-E3) :- B3=B1, E3=E2, B2=E1.
  \]

  Or, more compactly:

  \[
  \]

  And, actually, no call to \text{append_dl} is normally necessary!

  ... But can only be done once (see later).

• Also one can build difference (open ended) trees, dictionaries, queues, etc., by leaving variables at the ends (e.g., at the leaves for trees).
Playing with difference lists

- Create two difference lists \( L_1 \) and \( L_2 \) and append them \( L_2 = X \) “by hand”:

\[
\text{?- } L_1 = [1,2,3|X], \quad L_2 = [4,5|Y], \quad L_2 = X.
\]

\[
L_1 = [1,2,3,4,5|Y],
\]

\[
L_2 = [4,5|Y],
\]

\[
X = [4,5|Y] ?
\]

yes

\( L_1 \) contains the resulting difference list \([1,2,3,4,5|Y]\).

- Given: \( \text{append_dl}(B1-E1,E1-E2,B1-E2) \)

\[
? - \text{append_dl}([1,2,3|X]-X,[4,5|Y]-Y,L).
\]

\[
L = [1,2,3,4,5|Y]-Y, \quad X = [4,5|Y] ?
\]

\( L \) has the resulting (appended) difference list.
But note that we have modified the first list: we cannot append to it again.

\[
? - \text{append_dl}(L-X,[4,5|Y]-Y,[1,2,3,4,5|Z]-Z).
\]

\[
L = [1,2,3,4,5|Y], \quad X = [4,5|Y], \quad Z = Y ?
\]
Standard qsort (using append)

```prolog
qsort([], []).
qsort([X|L], R) :-
    partition(L, X, L1, L2),
    qsort(L2, R2),
    qsort(L1, R1),
    append(R1, [X|R2], R).
```

```prolog
partition([], _B, [], []).
partition([E|R], C, [E|Left1], Right) :-
    E < C,
    partition(R, C, Left1, Right).
```

```prolog
partition([E|R], C, Left, [E|Right1]) :-
    E >= C,
    partition(R, C, Left, Right1).
```
First list is normal list, second is built as a difference list.

\[
\text{dlqsort}(L, SL) :- \text{dlqsort}_1(L, SL, []). \\
\text{dlqsort}_1([], R, R). \\
\text{dlqsort}_1([X|L], R, R1) :- \\
\quad \text{partition}(L, X, L1, L2), \\
\quad \text{dlqsort}_1(L1, R, [X|R0]), \\
\quad \text{dlqsort}_1(L2, R0, R1).
\]

% Partition is the same as before.
Parsing (using append and traditional lists)

?- myphrase([t,h,e,’’,p,l,a,n,e,’’,f,l,i,e,s]).

myphrase(X) :-
    append(A,T1,X), article(A), append(SP,T2,T1),
    spaces(SP), append(N,T3,T2), noun(N),
    append(SPN,V,T3), spaces(SPN), verb(V).

article([a]).
article([t,h,e]).

spaces([’ ’]).
spaces([’ ’ | Y]) :- spaces(Y).

noun([c,a,r]).
noun([p,l,a,n,e]).

verb([f,l,i,e,s]).
verb([d,r,i,v,e,s]).
Parsing (using standard clauses and difference lists)

?- myphrase([t,h,e,’ ’,p,l,a,n,e,’ ’,f,l,i,e,s],[[]]).

myphrase(X,CV) :-
   article(X,CA), spaces(CA,CS1), noun(CS1,CN),
   spaces(CN,CS2), verb(CS2,CV).

article([a|X],X).
article([t,h,e|X],X).

spaces([ ’ ’ | X],X).
spaces([ ’ ’ | Y],X) :- spaces(Y,X).

noun([c,a,r | X],X).
noun([p,l,a,n,e | X],X).

verb([f,l,i,e,s | X],X).
verb([d,r,i,v,e,s | X],X).
Parsing (same, using some string syntax)

?- myphrase("the plane flies",[]).

myphrase(X,CV) :-
    article(X,CA), spaces(CA,CS1), noun(CS1,CN),
    spaces(CN,CS2), verb(CS2,CV).

article("a" || X, X).
article("the" || X, X).

spaces(" ") || X, X).
spaces(" ") || Y, X).

noun("car" || X, X).
noun("plane" || X, X).

verb("flies" || X, X).
verb("drives" || X, X).
Parsing (same, using additional syntax: DCGs)

- Add syntactic transformation to avoid writing all the auxiliary variables. The result is called a **Definite Clause Grammar** ("DCG").

```prolog
?- myphrase("the plane flies",[]). or, use the "phrase/2" builtin:
?- phrase(myphrase,"the plane flies").
```

```prolog
:- use_package(dcg).

myphrase --> article, spaces, noun, spaces, verb.

article --> "a".
article --> "the".
noun --> "car".
noun --> "plane".

 spaces --> " ".
 spaces --> " ", spaces.
 verb --> "flies".
 verb --> "drives".
```
Parsing + actions (calling Prolog in DCGs)

- Other actions can be interspersed with the grammar.
  Raw Prolog can be called (between "{...}")

?- myphrase(NChars,"the plane flies",[]).
?- phrase(myphrase(N),"the plane flies").

:- use_package(dcg).

myphrase(N) --> article(AC), spaces(S1), noun(NC), spaces(S2), verb(VC), { N is AC + S1 + NC + S2 + VC}.

article(1) --> "a".
article(3) --> "the".
spaces(1) --> " ".
spaces(N) --> " ", spaces(N1), { N is N1+1 }.
noun(3) --> "car".
noun(5) --> "plane".
verb(5) --> "flies".
verb(6) --> "drives".
• Repeat loops.

```
main(_) :- repeat, read(X), process(X).
```

```
process(end).
process(X) :- display(X), fail.
```

• Exception handling.

• Extending the syntax beyond operators: term expansions/macros → packages.

• Delay declarations/concurrency.

• Operating system interface (and sockets, etc.).

• Foreign language (e.g., C) interfaces.

• Many other built-ins...

• ...
Some Typical Libraries in Prolog Systems

- Most systems have a good set of libraries.
- Worth checking before re-implementing existing functionality!
- Some examples:

<table>
<thead>
<tr>
<th>Libraries</th>
<th>Arrays</th>
<th>Assoc</th>
<th>Attributes</th>
<th>Heaps</th>
</tr>
</thead>
<tbody>
<tr>
<td>Lists</td>
<td></td>
<td></td>
<td>Ordset</td>
<td>Queues</td>
</tr>
<tr>
<td>Random</td>
<td></td>
<td></td>
<td>Tree</td>
<td>UGraphs</td>
</tr>
<tr>
<td>WGraphs</td>
<td></td>
<td></td>
<td>Linda/Distribution</td>
<td>Persistent DB</td>
</tr>
<tr>
<td>CLPB</td>
<td></td>
<td>CLPQR</td>
<td>CLPFD</td>
<td>Objects</td>
</tr>
<tr>
<td>GCLA</td>
<td></td>
<td>TclTk</td>
<td>Tracing</td>
<td>Chars I/O</td>
</tr>
<tr>
<td>Runtime Utilities</td>
<td></td>
<td>Timeout</td>
<td>Xrefs</td>
<td>WWW</td>
</tr>
<tr>
<td>Java Interface</td>
<td></td>
<td>...</td>
<td>...</td>
<td>...</td>
</tr>
</tbody>
</table>
Some Additional Libraries and Extensions (Ciao)

Other systems may offer additional extensions. Some examples from Ciao:

- Other execution rules:
  - Breadth-first, Iterative-deepening, Random, ...
  - Tabling
  - CASP (negation with multiple models)
  - Andorra (“determinate-first”) execution
  - Fuzzy Prolog, MYCIN rules, ...

- Interfaces to other languages and systems:
  - Interfaces to C, Java, JavaScript, Python, LLVM, ...
  - SQL database interface and Persistent predicates
  - Web/HTML/XML/CGI programming (PiLLoW) / HTTP connectivity
  - Interfaces to wxWidgets, Tcl/Tk, VRML (ProVRML), ...
  - Interfaces to solvers (PPL, Mathematica, MiniSAT, ...)
  - Graphviz, daVinci interfaces
  - Calling Emacs from Prolog, etc.
• Many syntactic and semantic extensions:
  ◆ Functional notation
  ◆ Higher-order
  ◆ Terms with named arguments - records/feature terms
  ◆ Multiple argument indexing
  ◆ The script interpreter
  ◆ Active modules (high-level distributed execution)
  ◆ Concurrency/multithreading
  ◆ Attributed variables
  ◆ Object oriented programming
  ◆ ...

Some Additional Libraries and Extensions (Ciao, Contd.)
Some Additional Libraries and Extensions (Ciao, Contd.)

- Constraint programming (CLP)
  - rationals, reals, finite domains, ...
  - CHR (constraint handling rules), GeCode,

- Assertions:
  - Regular types, Modes, Determinacy, etc.
  - Other properties
  - Run-time checking of assertions
  - Assertion-based unit tests and automatic test case generation
  - Compile-time property inference and assertion checking (CiaoPP).

- Additional programming support:
  - Automatic documentation (LPdoc).
  - Partial evaluation, optimization, parallelization (CiaoPP).
  - ...